
AMath 483/583 — Lecture 24

Outline:

• Heat equation and discretization
• OpenMP and MPI for iterative methods
• Jacobi, Gauss-Seidel, SOR

Notes and Sample codes:

• Class notes: Linear algebra software

• $UWHPSC/codes/openmp/jacobi1d_omp1.f90

• $UWHPSC/codes/openmp/jacobi1d_omp2.f90

• $UWHPSC/codes/mpi/jacobi1d_mpi.f90

Steady state diffusion

If f(x, t) = f(x) does not depend on time and if the boundary
conditions don’t depend on time, then u(x, t) will converge
towards steady state distribution satisfying

0 = Duxx(x) + f(x)

(by setting ut = 0.)

This is now an ordinary differential equation (ODE) for u(x).

We can solve this on an interval, say 0 ≤ x ≤ 1 with

Boundary conditions:

u(0) = α, u(1) = β.

Finite difference method

Ui ≈ u(xi)

ux(xi+1/2) ≈ Ui+1−Ui

∆x

ux(xi−1/2) ≈ Ui−Ui−1

∆x

So we can approximate second derivative at xi by:

uxx(xi) ≈
1

∆x

(
Ui+1 − Ui

∆x
− Ui − Ui−1

∆x

)

=
1

∆x2
(Ui−1 − 2Ui + Ui+1)

This gives coupled system of n linear equations:

1

∆x2
(Ui−1 − 2Ui + Ui+1) = −f(xi)

for i = 1, 2, . . . , n. With U0 = α and Un+1 = β.

Iterative methods

Coupled system of n linear equations:

(Ui−1 − 2Ui + Ui+1) = −∆x2f(xi)

for i = 1, 2, . . . , n. With U0 = α and Un+1 = β.

Iterative method starts with initial guess U [0] to solution and
then improves U [k] to get U [k+1] for k = 0, 1, . . ..

Note: Generally does not involve modifying matrix A.

Do not have to store matrix A at all, only know about stencil.

Jacobi iteration

(Ui−1 − 2Ui + Ui+1) = −∆x2f(xi)

Solve for Ui:

Ui =
1

2

(
Ui−1 + Ui+1 + ∆x2f(xi)

)
.

Note: With no heat source, f(x) = 0,
the temperature at each point is average of neighbors.

Suppose U [k] is a approximation to solution. Set

U
[k+1]
i =

1

2

(
U

[k]
i−1 + U

[k]
i+1 + ∆x2f(xi)

)
for i = 1, 2, . . . , n.

Repeat for k = 0, 1, 2, . . . until convergence.

Can be shown to converge (eventually... very slow!)

Jacobi iteration in Fortran

uold = u ! starting values before updating

do iter=1,maxiter

dumax = 0.d0

do i=1,n
u(i) = 0.5d0*(uold(i-1) + uold(i+1) + dx**2*f(i))
dumax = max(dumax, abs(u(i)-uold(i)))
enddo

! check for convergence:
if (dumax .lt. tol) exit

uold = u ! for next iteration
enddo

Note: we must use old value at i− 1 for Jacobi.

Otherwise we get the Gauss-Seidel method.
u(i) = 0.5d0*(u(i-1) + u(i+1) + dx**2*f(i))

This actually converges faster!

Jacobi with OpenMP – coarse grain

General Approach:

• Fork threads only once at start of program.

• Each thread is responsible for some portion of the arrays,
from i=istart to i=iend.

• Each iteration, must copy u to uold, update u, check for
convergence.

• Convergence check requires coordination between threads
to get global dumax.

• Print out final result after leaving parallel block

See code in the repository or the notes:
$UWHPSC/codes/openmp/jacobi1d_omp2.f90

Jacobi with MPI

Each process is responsible for some portion of the arrays,
from i=istart to i=iend.

No shared memory: each process only has part of array.

Updating formula:
u(i) = 0.5d0*(uold(i-1) + uold(i+1) + dx**2*f(i))

Need to exchange values at boundaries:
Updating at i=istart requires uold(istart-1)
Updating at i=iend requires uold(istart+1)

Example with n = 9 interior points (plus boundaries):

Process 0 has istart = 1, iend = 5
Process 1 has istart = 6, iend = 9

Jacobi with MPI — Sending to neighbors

call mpi_comm_rank(MPI_COMM_WORLD, me, ierr)
...

do iter = 1, maxiter
uold = u

if (me > 0) then
! Send left endpoint value to "left"
call mpi_isend(uold(istart), 1, MPI_DOUBLE_PRECISION, &

me - 1, 1, MPI_COMM_WORLD, req1, ierr)
end if

if (me < ntasks-1) then
! Send right endpoint value to "right"
call mpi_isend(uold(iend), 1, MPI_DOUBLE_PRECISION, &

me + 1, 2, MPI_COMM_WORLD, req2, ierr)
end if

end do

Note: Non-blocking mpi_isend is used,

Different tags (1 and 2) for left-going, right-going messages.

R.J. LeVeque, University of Washington AMath 483/583, Lecture 24

Notes:

R.J. LeVeque, University of Washington AMath 483/583, Lecture 24



Jacobi with MPI — Receiving from neighbors

Note: uold(istart) from me+1 goes into uold(iend+1):
uold(iend) from me-1 goes into uold(istart-1):

do iter = 1, maxiter

! mpi_send’s from previous slide

if (me < ntasks-1) then
! Receive right endpoint value
call mpi_recv(uold(iend+1), 1, MPI_DOUBLE_PRECISION, &

me + 1, 1, MPI_COMM_WORLD, mpistatus, ierr)
end if

if (me > 0) then
! Receive left endpoint value
call mpi_recv(uold(istart-1), 1, MPI_DOUBLE_PRECISION, &

me - 1, 2, MPI_COMM_WORLD, mpistatus, ierr)
end if

! Apply Jacobi iteration on my section of array
do i = istart, iend

u(i) = 0.5d0*(uold(i-1) + uold(i+1) + dx**2*f(i))
dumax_task = max(dumax_task, abs(u(i) - uold(i)))
end do

end do

Jacobi with MPI

Other issues:

• Convergence check requires coordination between
processes to get global dumax.
Use MPI_ALLREDUCE so all process check same value.

• Part of final result must be printed by each process
(into common file heatsoln.txt), in proper order.

See code in the repository or the notes:
$UWHPSC/codes/mpi/jacobi1d_mpi.f90

Jacobi with MPI — Writing solution in order

Want to write table of values x(i),u(i) in heatsoln.txt.

Need them to be in proper order, so Process 0 must write to
this file first, then Process 1, etc.

Approach:

Each process me waits for a message from me-1 indicating that
it has finished writing its part. (Contents not important.)

Each process must open the file (without clobbering values
already there), write to it, then close the file.

Assumes all processes share a file system!

On cluster or supercomputer, need to either:
send all results to single process for writing, or
write distributed files that may need to be combined later

(some visualization tools handle distributed data!)

Jacobi in 2D

Updating point 7 for example (u32):

U
[k+1]
32 =

1

4
(U

[k]
22 + U

[k]
42 + U

[k]
21 + U

[k]
41 + h2f32)

Jacobi in 2D using MPI

With two processes: Could partition unknown into
Process 0 takes grid points 1–8
Process 1 takes grid points 9–16

Each time step:
Process 0 sends top boundary (5–8) to Process 1,
Process 1 sends bottom boundary (9–12) to Process 0.

Jacobi in 2D using MPI

With more grid points and processes...
Could partition several different ways, e.g. with 4 processes:

The partition on the right requires less communication.

With m2 processes on grid with n2 points:
2(m2 − 1)n boundary points on left,
4(m− 1)n boundary points on right.

Jacobi in 2D using MPI

For partition on left: Natural to number processes 0,1,2,3 and
pass boundary data from Process k to k ± 1.

For m×m array of processors as on right: How do we figure
out the neighboring process numbers?

Creating a communicator for Cartesian blocks

integer dims(2)
logical isperiodic(2), reorder

ndim = 2 ! 2d grid of processes
dims(1) = 4 ! for 4x6 grid of processes
dims(2) = 6
isperiodic(1) = .false. ! periodic in x?
isperiodic(2) = .false. ! periodic in y?
reorder = .true. ! optimize ordering

call MPI_CART_CREATE(MPI_COMM_WORLD, ndim, &
dims, isperiodic, reorder, comm2d, ierr)

Create communicator comm2d. See also:

MPI_CART_CREATE, MPI_CART_SHIFT, MPI_CART_COORDS.

Gauss-Seidel iteration in Fortran

do iter=1,maxiter

dumax = 0.d0

do i=1,n
uold = u(i)
u(i) = 0.5d0*(u(i-1) + u(i+1) + dx**2*f(i))
dumax = max(dumax, abs(u(i)-uold))
enddo

! check for convergence:
if (dumax .lt. tol) exit

enddo

Note: Now u(i) depends on value of u(i-1) that has already
been updated for previous i.

Good news: This converges about twice as fast as Jacobi!

But... loop carried dependence! Cannot parallelize so easily.

Red-black ordering

We are free to write equations of linear system in any order...
reordering rows of coefficient matrix, right hand side.

Can also number unknowns of linear system in any order...
reordering elements of solution vector.

Red-black ordering: Iterate through points with odd index first
(i = 1, 3, 5, . . .) and then even index points (i = 2, 4, 6, . . .).

Then all black points can be updated in any order,
all red points can then be updated in any order.

Same asymptotic convergence rate as natural ordering.

Red-Black Gauss-Seidel

do iter=1,maxiter
dumax = 0.d0

! UPDATE ODD INDEX POINTS:
!$omp parallel do reduction(max : dumax) &
!$omp private(uold)
do i=1,n,2

uold = u(i)
u(i) = 0.5d0*(u(i-1) + u(i+1) + dx**2*f(i))
dumax = max(dumax, abs(u(i)-uold))
enddo

! UPDATE EVEN INDEX POINTS:
!$omp parallel do reduction(max : dumax) &
!$omp private(uold)
do i=2,n,2

uold = u(i)
u(i) = 0.5d0*(u(i-1) + u(i+1) + dx**2*f(i))
dumax = max(dumax, abs(u(i)-uold))
enddo

! check for convergence:
if (dumax .lt. tol) exit
enddo

Gauss-Seidel method in 2D

If ∆x = ∆y = h:

1

h2
(Ui−1,j + Ui+1,j + Ui,j−1 + Ui,j+1 − 4Ui,j) = −f(xi, yj).

Solve for Ui,j and iterate:

u
[k+1]
i,j =

1

4
(u

[k+1]
i−1,j + u

[k]
i+1,j + u

[k+1]
i,j−1 + u

[k]
i,j+1 − h2fi,j)

Again no need for matrix A.

Note: Above indices for old and new values assumes we iterate
in the natural row-wise order.

R.J. LeVeque, University of Washington AMath 483/583, Lecture 24

Notes:

R.J. LeVeque, University of Washington AMath 483/583, Lecture 24



Gauss-Seidel in 2D

Updating point 7 for example (u32):

Depends on new values at points 6 and 3, old values at points
7 and 10.

U
[k+1]
32 =

1

4
(U

[k+1]
22 + U

[k]
42 + U

[k+1]
21 + U

[k]
41 + h2f32)

Red-black ordering in 2D

Again all black points can be updated in any order:
New value depends only on red neighbors.

Then all red points can be updated in any order:
New value depends only on black neighbors.

SOR method

Gauss-Seidel move solution in right direction but not far enough
in general.

Iterates “relax” towards solution.

Successive Over-Relaxation (SOR):

Compute Gauss-Seidel approximation and then go further:

U∗
i =

1

2
(U

[k+1]
i−1 + U

[k]
i+1 + ∆x2f(xi))

U
[k+1]
i = U

[k]
i + ω(U∗

i − U
[k]
i )

where 1 < ω < 2.

Optimal omega (For this problem): ω = 2 − 2π∆x.

Convergence rates

Red-Black SOR in 1D

do iter=1,maxiter
dumax = 0.d0

! UPDATE ODD INDEX POINTS:
!$omp parallel do reduction(max : dumax) &
!$omp private(uold, ustar)
do i=1,n,2

uold = u(i)
ustar = 0.5d0*(u(i-1) + u(i+1) + dx**2*f(i))
u(i) = uold + omega*(ustar-uold)
dumax = max(dumax, abs(u(i)-uold))
enddo

! UPDATE EVEN INDEX POINTS:
!$omp parallel do reduction(max : dumax) &
!$omp private(uold, ustar)
do i=2,n,2

uold = u(i)
ustar = 0.5d0*(u(i-1) + u(i+1) + dx**2*f(i))
u(i) = uold + omega*(ustar-uold)
dumax = max(dumax, abs(u(i)-uold))
enddo

! check for convergence...

Note that uold, ustar must be private!

Announcements

Homework 6 is in the notes and due next Friday.

Quizzes for this week’s lectures due next Wednesday.

Office hours today 9:30 – 10:20.

Next week:

Monday: no class

Wednesday: Guest lecture —

Brad Chamberlain, Cray

Chapel: A Next-Generation Partitioned
Global Address Space (PGAS) Language

R.J. LeVeque, University of Washington AMath 483/583, Lecture 24

Notes:

R.J. LeVeque, University of Washington AMath 483/583, Lecture 24


